5.1 Rovnice shodnosti v rovině

Každou afinitu f v rovině můžeme zapsat soustavou rovnic

$$\begin{aligned} f:\ x' &= a_{11}x + a_{12}y + b_1 \\ y' &= a_{21}x + a_{22}y + b_2, \end{aligned} \tag{21}$$

kterou přepíšeme užitím matic do tvaru

$$f:\ \begin{bmatrix} x' \\ y' \end{bmatrix} = \begin{bmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{bmatrix} \cdot \begin{bmatrix} x \\ y \end{bmatrix} + \begin{bmatrix} b_1 \\ b_2 \end{bmatrix} \tag{22}$$

a stručně vyjádříme rovnicí

$$f:\ X' = A \cdot X + B. \tag{23}$$

Jak poznáme, že afinita (21) je shodností?

Je-li tato afinita shodností, platí pro všechny dvojice bodů $X[x_1, x_2], Y[y_1, y_2]$ a jejich obrazy $X'[x_1', x_2'], Y'[y_1', y_2']$ vztah $|X'Y'| = |XY|$, z něhož po dosazení souřadnic uvedených bodů dostaneme

$$\sqrt{(y_1' - x_1')^2 + (y_2' - x_2')^2} = \sqrt{(y_1 - x_1)^2 + (y_2 - x_2)^2}, \tag{24}$$

po umocnění obou stran na druhou

$$(y_1' - x_1')^2 + (y_2' - x_2')^2 = (y_1 - x_1)^2 + (y_2 - x_2)^2. \tag{25}$$

Nyní do levé strany (25) dosadíme z (21), upravíme na tvar obsahující výrazy $(y_1 - x_1)$ a $(y_2 - x_2)$ a diskutujeme, za jakých podmínek je splněna její rovnost s pravou stranou. Zjistíme, že rovnost $|X'Y'| = |XY|$ nastává právě tehdy, když jsou pro prvky matice A (tj. koeficienty soustavy (21)) splněny vztahy

$$\begin{aligned} a_{11}^2 + a_{21}^2 &= 1, \\ a_{12}^2 + a_{22}^2 &= 1, \\ a_{11}a_{12} + a_{21}a_{22} &= 0, \end{aligned} \tag{26}$$

které lze stručně vyjádřit rovností

$$\begin{bmatrix} a_{11} & a_{21} \\ a_{12} & a_{22} \end{bmatrix} \cdot \begin{bmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{bmatrix} = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}. \tag{27}$$

Odpověď na výše uvedenou otázku je tedy taková, že **rovnice** (21) **je rovnicí shodnosti, právě když platí**

$$A^T \cdot A = E, \tag{28}$$

kde E **je jednotková matice,** jinak řečeno, když je matice A **ortonormální**.

Poznámky.

1. Platí $A^T \cdot A = E$. Potom je ale $A^T = A^{-1}$ a platí tedy i rovnost $A \cdot A^T = E$.

2. Zobrazení, pro která platí $|\det A| = 1$ nazýváme ekviafinní zobrazení, stručně **ekviafinity**. Je zřejmé, že každá shodnost je ekviafinita. Platí toto tvrzení i obráceně? Můžeme říci, že každá ekviafinita je shodností?

3. Je třeba si uvědomit, že při shodném zobrazení mezi euklidovskými prostory různých dimenzí není matice A čtvercová. Potom výše uvedené úvahy o inverzní matici nemají smysl a v platnosti zůstává pouze původní podmínka $A^T \cdot A = E$.

Samodružné body

Samodružným bodem (afinního) zobrazení rozumíme bod, který se zobrazí sám na sebe, tj. pro jeho souřadnice platí $X' = X$. Pokud do rovnic (21) dosadíme $x' = x$ a $y' = y$ je zřejmé, že souřadnice samodružných bodů dané shodnosti jsou řešením soustavy rovnic

$$\begin{aligned} (1 - a_{11})x_1 - a_{12}x_2 &= b_1 \\ -a_{21}x_1 + (1 - a_{22})x_2 &= b_2. \end{aligned} \tag{29}$$

Samodružné směry

Samodružným směrem rozumíme směr, který se v (afinním) zobrazení zobrazí sám na sebe. Pro vyjádření směru používáme vektor, např. $\vec{u}$ (příslušný „směr“ potom reprezentují všechny jeho násobky). Má-li být tento směr samodružný, musí pro vektor $\vec{u}'$, který je obrazem vektoru $\vec{u}$, platit $\vec{u}' = \lambda\vec{u}$, kde $\lambda \in \mathbb{R}$.

Zobrazení mezi vektory zaměření afinního bodového prostoru (obecně však toto zobrazení probíhá mezi různými zaměřeními různých bodových prostorů) zajišťuje tzv. **asociovaný homomorfismus** (též **lineární zobrazení**).

Definice 16 (Homomorfismus)**.** *Zobrazení φ vektorového prostoru V do vektorového prostoru V' se nazývá homomorfismus (lineární zobrazení), jestliže pro všechna $\vec{u}, \vec{v} \in V$, $k \in \mathbb{T}$ (místo obecného tělesa $\mathbb{T}$ můžeme uvažovat $\mathbb{R}$) platí:*

$$\begin{aligned} &(1) \quad \varphi(\vec{u} + \vec{v}) = \varphi(\vec{u}) + \varphi(\vec{v}), \\ &(2) \quad \varphi(k\vec{u}) = k\varphi(\vec{u}). \end{aligned}$$

Definice 17 (Asociovaný homomorfismus zobrazení f v rovině)**.** *Uvažujme afinní transformaci f prostoru E_2. Potom* **asociovaným** *(tj. jednoznačně přiřazeným)* **homomorfismem** *afinity f rozumíme lineární zobrazení φ, které zobrazuje zaměření V_2 prostoru E_2 do sebe takto:*

$$\vec{u} = Y - X \Rightarrow \varphi(\vec{u}) = f(Y) - f(X), \tag{30}$$

kde X, Y a $f(X), f(Y)$ jsou body z E_2, $\vec{u}, \varphi(\vec{u}) \in V_2$.

Asociovaný homomorfismus φ afinity f je potom dán soustavou

$$\begin{aligned} \varphi:\ u_1' &= a_{11}u_1 + a_{12}u_2 \\ u_2' &= a_{21}u_1 + a_{22}u_2, \end{aligned}$$

maticově pak

$$\varphi: \begin{bmatrix} u_1' \\ u_2' \end{bmatrix} = \begin{bmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{bmatrix} \cdot \begin{bmatrix} u_1 \\ u_2 \end{bmatrix},$$

což lze zapsat, analogicky s rovnicí (22), ve tvaru

$$\varphi:\ \vec{u}' = A \cdot \vec{u}. \tag{31}$$

Samodružné směry shodnosti (tj. vektory těchto směrů, pro které platí $\vec{u}' = \lambda\vec{u}$) jsou potom **netriviálním** řešením homogenní soustavy rovnic

$$\begin{aligned} (\lambda - a_{11})u_1 - a_{12}u_2 &= 0 \\ -a_{21}u_1 + (\lambda - a_{22})u_2 &= 0. \end{aligned} \tag{32}$$

Homogenní soustava n lineárních rovnic o n neznámých má netriviální řešení právě tehdy, když je determinant soustavy roven nule. Soustavy (32) má tedy nekonečně mnoho řešení, jestliže platí rovnost

$$\begin{vmatrix} (\lambda - a_{11}) & -a_{12} \\ -a_{21} & (\lambda - a_{22}) \end{vmatrix} = 0. \tag{33}$$

Rovnici (33) říkáme **charakteristická rovnice** příslušného zobrazení, v tomto případě shodnosti v rovině. Každý vektor $\vec{u}$, pro který platí $\vec{u}' = \varphi(\vec{u}) = \lambda\vec{u}$, nazýváme **vlastním vektorem** homomorfismu φ, číslo λ, které je řešením charakteristické rovnice, pak nazýváme **vlastní číslo** homomorfismu φ, odpovídající vektoru $\vec{u}$. Místo vlastní vektor a vlastní číslo se také používají termíny **charakteristický vektor** a **charakteristické číslo**.

Uvedené postupy určení samodružných bodů a směrů shodného zobrazení si nyní budeme ilustrovat na nám dobře známých shodnostech, na středové a osové souměrnosti.

Středová souměrnost se středem v bodě $S = [2, -3]$ je dána rovnicemi

$$x' = -x + 4,$$
$$y' = -y - 6.$$

Představme si, že nevíme, o jaké afinní zobrazení se jedná a teprve to chceme zjistit. Matice tohoto zobrazení je $A = \begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$, součin $A^T \cdot A$ je roven $A^T \cdot A = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$, jedná se tedy o shodnost.

Nyní určíme samodružné body daného zobrazení řešením soustavy

$$2x = 4,$$
$$2y = -6.$$

Ta má jediné řešení $[x, y] = [2, -3]$. Jedná se tedy o shodné zobrazení s jediným samodružným bodem $S = [2, -3]$. V úvahu tak připadá otočení nebo středová souměrnost.

K rozhodnutí, která z těchto dvou možností je správná, nám pomůže určení samodružných směrů daného zobrazení. Řešíme proto homogenní soustavu

$$(\lambda + 1)u_1 = 0,$$
$$(\lambda + 1)u_2 = 0,$$

které přísluší charakteristická rovnice

$$\begin{vmatrix} (\lambda + 1) & 0 \\ 0 & (\lambda + 1) \end{vmatrix} = 0,$$

po úpravě ve tvaru

$$(\lambda + 1)^2 = 0.$$

Jejím jediným řešením je vlastní číslo $\lambda = -1$, které dosadíme do příslušné homogenní soustavy, abychom dostali soustavu rovnic

$$0u_1 = 0,$$
$$0u_2 = 0,$$

jejímž řešením je každý vektor $\vec{v} = (u_1, u_2) \in R \times R$. Vyšetřovaná shodnost má tedy všechny směry samodružné. Jedná se proto o středovou souměrnost se středem $S = [2, -3]$.

Osová souměrnost s osou v souřadnicové ose x je dána rovnicemi

$$\begin{aligned} x' &= x, \\ y' &= -y. \end{aligned}$$

Opět předstíráme, že nevíme, o jaké afinní zobrazení se jedná a teprve to chceme zjistit.

Matice tohoto zobrazení je $A = \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}$, součin $A^T \cdot A$ je roven $A^T \cdot A = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$, jedná se tedy o shodnost.

Nyní určíme samodružné body daného zobrazení řešením soustavy

$$\begin{aligned} 0x &= 0, \\ 2y &= 0. \end{aligned}$$

Ta má nekonečně mnoho řešení. Jsou jimi všechny uspořádané dvojice ve tvaru $[x, 0]$; $x \in R$. Jedná se tedy o shodné zobrazení, jehož všechny samodružné body leží v přímce o rovnici $y = 0$. V úvahu tak připadá jediná možnost, osová souměrnost s osou v souřadnicové ose x.

Přestože jsme dané zobrazení již identifikovali, dokončíme analýzu jeho vlastností určením samodružných směrů. Řešíme proto homogenní soustavu

$$\begin{aligned} (\lambda - 1)u_1 &= 0, \\ (\lambda + 1)u_2 &= 0, \end{aligned}$$

které přísluší charakteristická rovnice

$$\begin{vmatrix} (\lambda - 1) & 0 \\ 0 & (\lambda + 1) \end{vmatrix} = 0,$$

po úpravě ve tvaru

$$(\lambda - 1)(\lambda + 1) = 0.$$

Charakteristická rovnice má dva kořeny (vlastní čísla) $\lambda_1 = 1, \lambda_2 = -1$, které postupně dosadíme do příslušné homogenní soustavy a vypočítáme souřadnice příslušných vlastních vektorů daného zobrazení.

Pro $\lambda_1 = 1$ dostáváme soustavu

$$\begin{aligned} 0u_1 &= 0, \\ 2u_2 &= 0, \end{aligned}$$

jejímž řešením je každý vektor $\vec{v}_1 = (u_1, 0) \in R^2$. Samodružný směr určený těmito vektory je rovnoběžný s osou x (tj. s osou souměrnosti).

Pro $\lambda_2 = -1$ dostáváme soustavu

$$\begin{aligned} -2u_1 &= 0, \\ 0u_2 &= 0, \end{aligned}$$

jejímž řešením je každý vektor $\vec{v}_2 = (0, u_2) \in R^2$. Samodružný směr určený těmito vektory je kolmý k ose x (tj. k ose souměrnosti). Určení dvou na sebe kolmých samodružných směrů je v souladu se skutečností, že uvažované shodné zobrazení je osová souměrnost.

PŘÍKLAD 5.2. *Zjistěte, zda existuje shodnost E_2, při které se bod $K = [10; 0]$ zobrazí na počátek $K' = [0; 0]$ a bod $L = [25; 20]$ na bod $L' = [0; 25]$. V kladném případě napište rovnice tohoto zobrazení a najděte jeho samodružné body a směry.*

Řešení: Začneme tím, že si ověříme, zda zadané body splňují definici shodného zobrazení, tj. zda $|K'L'| = |KL|$. V případě této úlohy zvládneme ověření provést zpaměti. Výsledkem je, že zadání vyhovuje definici shodnosti.

Další postup řešení úlohy si ilustrujeme pomocí zápisu v programu wxMaxima (viz http://andrejv.github.io/wxmaxima/)

```
(%i1) A:matrix([a11,a12],[a21,a22]); B:matrix([b1],[b2]);
```

(%o1) $\begin{pmatrix} a11 & a12 \\ a21 & a22 \end{pmatrix}$

(%o2) $\begin{pmatrix} b1 \\ b2 \end{pmatrix}$

Rovnici $X' = A \cdot X + B$ vyjádříme ve tvaru $A \cdot X + B - X' = O$ a dosadíme souřadnice daných dvojic bodů K, K' a L, L'. Potom zapíšeme podmínku (28) pro to, aby bylo afinní zobrazení shodností ve tvaru $A^T \cdot A - E = O$. (V programu wxMaxima zapíšeme jenom levé strany uvedených rovnic.)

```
(%i3) s1:A.[10,0]+B-[0,0]; s2:A.[25,20]+B-[0,25];
      s3:transpose(A).A-ident(2);
```

(%o3) $\begin{pmatrix} b1+10\,a11 \\ b2+10\,a21 \end{pmatrix}$

(%o4) $\begin{pmatrix} b1+20\,a12+25\,a11 \\ b2+20\,a22+25\,a21-25 \end{pmatrix}$

(%o5) $\begin{pmatrix} a21^2+a11^2-1 & a21\,a22+a11\,a12 \\ a21\,a22+a11\,a12 & a22^2+a12^2-1 \end{pmatrix}$

Všechny prvky výše uvedených matic musí být rovny nule (Proč?). Dostaneme tak soustavu sedmi rovnic pro šest neznámých $a_{11}, a_{12}, a_{21}, a_{22}, b_1, b_2$.

```
(%i6) rov:[s1[1,1],s1[2,1],s2[1,1],s2[2,1],s3[1,1],s3[1,2],s3[2,2]];
```

(%o6) $[b1+10\,a11, b2+10\,a21, b1+20\,a12+25\,a11, b2+20\,a22+25\,a21-25, a21^2+a11^2-1, a21\,a22+a11\,a12, a22^2+a12^2-1]$

Tato soustava má následující dvě řešení (nejedná se o soustavu lineárních rovnic, proto může mít dvě řešení):

```
(%i7) res:solve(rov,[a11,a12,a21,a22,b1,b2]);
```

(%o7) $[[a11=\frac{4}{5}, a12=-\frac{3}{5}, a21=\frac{3}{5}, a22=\frac{4}{5}, b1=-8, b2=-6], [a11=-\frac{4}{5}, a12=\frac{3}{5}, a21=\frac{3}{5}, a22=\frac{4}{5}, b1=8, b2=-6]]$

Dvěma řešením odpovídají dvě různé shodnosti. Zjistili jsme tedy, že existují dvě shodnosti, které převádějí body K, L na body K', L' (Což se, vzhledem ke *větě o určenosti shodného (afinního) zobrazení* dalo čekat. Proč?). Pokračujeme v řešení úlohy pro každou z těchto shodností zvlášť. Pro zápis rovnic uvažovaných shodností si nejprve připravíme matici `RovTr`, jejímiž řádky jsou rovnice afinity v obecném tvaru (tato matice není nutnou součástí postupu řešení, jedná se jenom o usnadnění vizuální prezentace rovnic v programu).

```
(%i8) RovTr:matrix([x1=a11*x+a12*y+b1],[y1=a21*x+a22*y+b2]);
```

(%o8) $\begin{pmatrix} x1=a12\,y+a11\,x+b1 \\ y1=a22\,y+a21\,x+b2 \end{pmatrix}$

Řešení č. 1:

```
(%i9)  A1:ev(A,res[1]); B1:ev(B,res[1]);
```

(%o9) $\begin{pmatrix} \frac{4}{5} & -\frac{3}{5} \\ \frac{3}{5} & \frac{4}{5} \end{pmatrix}$

(%o10) $\begin{pmatrix} -8 \\ -6 \end{pmatrix}$

Příslušná shodnost má rovnice

```
(%i11) R1:ev(RovTr,res[1]);
```

(%o11) $\begin{pmatrix} x1 = -\frac{3\,y}{5} + \frac{4\,x}{5} - 8 \\ y1 = \frac{4\,y}{5} + \frac{3\,x}{5} - 6 \end{pmatrix}$

Samodružný bod je bod, pro který platí $X' = X$. Pro výpočet souřadnic samodružných bodů daného zobrazení tak do rovnice $X' = A \cdot X + B$ (pro snazší zpracování programem přepsané do tvaru $A \cdot X + B - X = 0$) za X' dosadíme X a řešíme odpovídající soustavu dvou rovnic s neznámými x, y.

```
(%i12) RovSB1:A1.[x,y]+B1-[x,y]; solve([RovSB1[1,1],RovSB1[2,1]],[x,y]);
```

(%o12) $\begin{pmatrix} -\frac{3\,y}{5} - \frac{x}{5} - 8 \\ -\frac{y}{5} + \frac{3\,x}{5} - 6 \end{pmatrix}$

(%o13) $[[x = 5, y = -15]]$

Protože tato soustava má jediné řešení, má daná shodnost jediný samodružný bod $S = [5, -15]$.

Pro vyšetření samodružných směrů daného zobrazení řešíme charakteristickou rovnici (33)

```
(%i14) CharM1:A1-%lambda*ident(2);
       CharR1:expand(determinant(CharM1))=0;
       solve(CharR1,%lambda);
```

(%o14) $\begin{pmatrix} \frac{4}{5} - \lambda & -\frac{3}{5} \\ \frac{3}{5} & \frac{4}{5} - \lambda \end{pmatrix}$

(%o15) $\lambda^2 - \frac{8\,\lambda}{5} + 1 = 0$

(%o16) $[\lambda = -\frac{3\,i-4}{5}, \lambda = \frac{3\,i+4}{5}]$

Charakteristická rovnice nemá řešení v oboru reálných čísel. Daná shodnost tak nemá žádný samodružný směr.

Protože uvažované zobrazení má právě jeden samodružný bod a nemá žádný samodružný směr, jedná se o **otočení** se středem $S = [5, -15]$.

Poznámka. K úplné identifikaci daného zobrazení nám zbývá určit úhel otočení α. Jak to uděláme?

Řešení č. 2:

Postupujeme analogicky s řešením č. 1.

```
(%i17) A2:ev(A,res[2]); B2:ev(B,res[2]);
```

(%o17) $\begin{pmatrix} -\frac{4}{5} & \frac{3}{5} \\ \frac{3}{5} & \frac{4}{5} \end{pmatrix}$

(%o18) $\begin{pmatrix} 8 \\ -6 \end{pmatrix}$

Rovnice zobrazení

```
(%i19) R2:ev(RovTr,res[2]);
```

(%o19) $\begin{pmatrix} x1 = \frac{3\,y}{5} - \frac{4\,x}{5} + 8 \\ y1 = \frac{4\,y}{5} + \frac{3\,x}{5} - 6 \end{pmatrix}$

Samodružné body:

```
(%i20) RovSB2:A2.[x,y]+B2-[x,y]; solve([RovSB2[1,1],RovSB2[2,1]],[x,y]);
```

(%o20) $\begin{pmatrix} \frac{3\,y}{5} - \frac{9\,x}{5} + 8 \\ -\frac{y}{5} + \frac{3\,x}{5} - 6 \end{pmatrix}$

(%o21) $[]$

Toto zobrazení tedy nemá žádný samodružný bod.

Samodružné směry:

```
(%i22) CharM2:A2-%lambda*ident(2);
       CharR2:expand(determinant(CharM2))=0;
       solve(CharR2,%lambda);
```

(%o22) $\begin{pmatrix} -\lambda-\frac{4}{5} & \frac{3}{5} \\ \frac{3}{5} & \frac{4}{5}-\lambda \end{pmatrix}$

(%o23) $\lambda^2-1=0$

(%o24) $[\lambda=-1, \lambda=1]$

```
(%i25) RovSS2:A2.[u,v]-[%lambda*u,%lambda*v];
```

(%o25) $\begin{pmatrix} \frac{3\,v}{5}-\lambda\,u-\frac{4\,u}{5} \\ -\lambda\,v+\frac{4\,v}{5}+\frac{3\,u}{5} \end{pmatrix}$

```
(%i26) RovSS21:ev(RovSS2,%lambda=-1);
       solve([RovSS21[1,1],RovSS21[2,1]],[u,v]);
```

(%o26) $\begin{pmatrix} \frac{3\,v}{5}+\frac{u}{5} \\ \frac{9\,v}{5}+\frac{3\,u}{5} \end{pmatrix}$ $solve: dependentequationseliminated: (2)$

(%o27) $[[u=-3\,\%r1, v=\%r1]]$

```
(%i28) RovSS22:ev(RovSS2,%lambda=1);
       solve([RovSS22[1,1],RovSS22[2,1]],[u,v]);
```

(%o28) $\begin{pmatrix} \frac{3\,v}{5}-\frac{9\,u}{5} \\ \frac{3\,u}{5}-\frac{v}{5} \end{pmatrix}$ $solve: dependentequationseliminated: (2)$

(%o29) $[[u=\frac{\%r2}{3}, v=\%r2]]$

Zobrazení má dva na sebe kolmé samodružné směry $\vec{u}=(-3,1), \vec{u}=(1,3)$.

Jedná se o **posunuté zrcadlení**.

Poznámka. K úplné identifikaci výsledného zobrazení nám zbývá určit osu o a vektor posunutí $\vec{t}$. Jak to uděláme?